```
Clear[β, γ, λ]
(*The model is -- eqn 12-14 in Bjornstad et al.
     N0 (t) = Exp (ξ) S (t)  => N0 (t) = Exp (ξ) (λ N1 (t-1) + λ N2 (t-1) + λ N3 (t-1))
   N1 (t) = N0 (t-1) Exp[- β Log[N0 (t-1)] - β Log[N1 (t-1)]]
    N2 (t) = λ N1 (t-1)
     N3 (t) = λ N2 (t-1)
      N4 (t) = λ N3 (t-1)
       S (t) = N2 (t)+N3 (t)+N4 (t) = λ N1 (t-1) + λ N2 (t-1) + λ N3 (t-1)
 *)

(*Let x0 = Log[N0], x1 = Log[N1], etc
   The following is then the model re-written in log-abundances: *)
f0 = Exp[ξ] (λ Exp[x1] + λ Exp[ x2] + λ Exp[x3]);
f1 = Exp[x0] Exp[-β x0 - γ x1];
f2 = λ Exp[x1];
f3 = λ Exp[x2];
(*f4=λ Exp[x3] is not needed*)
fs = λ Exp[x1] + λ Exp[ x2] + λ Exp[x3];

(* Differentiating the log-model with respect to log-abundances (see eqn A4): *)
j11 = D[Log[f0], x0]; j12 = D[Log[f0], x1];
j13 = D[Log[f0], x2]; j14 = D[Log[f0], x3]; j15 = D[Log[f0], s];
j21 = D[Log[f1], x0]; j22 = D[Log[f1], x1];
j23 = D[Log[f1], x2]; j24 = D[Log[f1], x3]; j25 = D[Log[f1], s];
j31 = D[Log[f2], x0]; j32 = D[Log[f2], x1];
j33 = D[Log[f2], x2]; j34 = D[Log[f2], x3]; j35 = D[Log[f2], s];
j41 = D[Log[f3], x0]; j42 = D[Log[f3], x1];
j43 = D[Log[f3], x2]; j44 = D[Log[f3], x3]; j45 = D[Log[f3], s];
j51 = D[Log[fs], x0] ; j52 = D[Log[fs], x1] ;
j53 = D[Log[fs], x2] ; j54 = D[Log[fs], x3] ; j55 = D[Log[fs], s] ;


(* Defining the Jacobian for the log-linearized system *)
J = {{j11, j12, j13, j14, j15}, {j21, j22, j23, j24, j25},
   {j31, j32, j33, j34, j35}, {j41, j42, j43, j44, j45}, {j51, j52, j53, j54, j55}};
MatrixForm[J]
```

$$\begin{pmatrix} 0 & \frac{e^{x1}\lambda}{e^{x1}\lambda+e^{x2}\lambda+e^{x3}\lambda} & \frac{e^{x2}\lambda}{e^{x1}\lambda+e^{x2}\lambda+e^{x3}\lambda} & \frac{e^{x3}\lambda}{e^{x1}\lambda+e^{x2}\lambda+e^{x3}\lambda} & 0 \\ 1-\beta & -\gamma & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & \frac{e^{x1}\lambda}{e^{x1}\lambda+e^{x2}\lambda+e^{x3}\lambda} & \frac{e^{x2}\lambda}{e^{x1}\lambda+e^{x2}\lambda+e^{x3}\lambda} & \frac{e^{x3}\lambda}{e^{x1}\lambda+e^{x2}\lambda+e^{x3}\lambda} & 0 \end{pmatrix}$$

```
(* simplifying at equilibrium *)
Jeq = Simplify[J /. Exp[x3] → λ Exp[x2] /. Exp[x2] → λ Exp[x1]];
MatrixForm[Jeq]
```

$$\begin{pmatrix} 0 & \frac{1}{1+\lambda+\lambda^2} & \frac{\lambda}{1+\lambda+\lambda^2} & \frac{\lambda^2}{1+\lambda+\lambda^2} & 0 \\ 1-\beta & -\gamma & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & \frac{1}{1+\lambda+\lambda^2} & \frac{\lambda}{1+\lambda+\lambda^2} & \frac{\lambda^2}{1+\lambda+\lambda^2} & 0 \end{pmatrix}$$

```
(* Defining the A vector (cf eqn A5) *)
a1 = D[Log[f0], ξ]; a2 = D[Log[f1], ξ];
a3 = D[Log[f3], ξ]; a4 = D[Log[f4], ξ]; a5 = D[Log[fs], ξ];
A = Transpose[{{a1, a2, a3, a4, a5}}];
MatrixForm[A]
```

$$\begin{pmatrix} 1 \\ 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}$$

```
(*Calculating the vector of transfer functions -- eqn 5 in Bjornstad et al.
   NOTE: the below is cast in terms of angular frequencies, w: w = 2 π f*)
Id := IdentityMatrix[5];
F := Inverse [Id - Exp[-I w] Jeq] .A
MatrixForm[F]
```

$$\begin{pmatrix} \frac{1+e^{-i w}\gamma}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}} \\ \frac{e^{-i w}-e^{-i w}\beta}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}} \\ \frac{e^{-2 i w}-e^{-2 i w}\beta}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}} \\ \frac{e^{-3 i w}-e^{-3 i w}\beta}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}} \\ -\frac{e^{-i w}(1-\beta)\left(-\frac{e^{-i w}}{1+\lambda+\lambda^2}-\frac{e^{-2 i w}\lambda}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda^2}{1+\lambda+\lambda^2}\right)}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}} \end{pmatrix}$$

```
T := ExpToTrig[F]
PS := Re[T]^2 + Im[T]^2;

(*The transfer function for age 0 cohort*)
F[[1]]
```

$$\left\{\frac{1+e^{-i w}\gamma}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}}\right\}$$

```
(*delay coordinate AR-coefficients for young-of-the-year*)
Simplify[Coefficient[- Denominator[F[[1]]][[1]], {e^(-i w), e^(-2 i w), e^(-3 i w), e^(-4 i w)}]]
(*These are the x0 (t-1), ..., x0 (t-4) coefficients*)
```

$$\left\{-\gamma, \frac{1-\beta}{1+\lambda+\lambda^2}, \frac{\lambda-\beta\lambda}{1+\lambda+\lambda^2}, -\frac{(-1+\beta)\lambda^2}{1+\lambda+\lambda^2}\right\}$$

```
(*AND delay coordinate MA-coeficients for young-of-the-year*)
Simplify[Coefficient[Numerator[F[[1]]][[1]], {e^(-i w), e^(-2 i w), e^(-3 i w), e^(-4 i w)}]]
(*These are the α (t), ..., α (t-3) coeficients*)
```

$$\{\gamma, 0, 0, 0\}$$

```
(*The transfer function for age 1 cohort*)
F[[2]]
```

$$\left\{\frac{e^{-i w}-e^{-i w}\beta}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}}\right\}$$

```
(*The transfer function for spawning stock cohort*)
F[[5]]
```

$$\left\{-\frac{e^{-i w}(1-\beta)\left(-\frac{e^{-i w}}{1+\lambda+\lambda^2}-\frac{e^{-2 i w}\lambda}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda^2}{1+\lambda+\lambda^2}\right)}{1+e^{-i w}\gamma-\frac{e^{-2 i w}}{1+\lambda+\lambda^2}+\frac{e^{-2 i w}\beta}{1+\lambda+\lambda^2}-\frac{e^{-3 i w}\lambda}{1+\lambda+\lambda^2}+\frac{e^{-3 i w}\beta\lambda}{1+\lambda+\lambda^2}-\frac{e^{-4 i w}\lambda^2}{1+\lambda+\lambda^2}+\frac{e^{-4 i w}\beta\lambda^2}{1+\lambda+\lambda^2}}\right\}$$

```
(*delay coordinate AR-coefficients for stock*)
Simplify[Coefficient[Expand[-Denominator[F[[5]]][[1]]], {e^(-i w), e^(-2 i w), e^(-3 i w), e^(-4 i w)}]]
(*These are the s (t-1), ..., s (t-4) coefficients*)
```

$$\left\{\frac{1-\beta}{1+\lambda+\lambda^2}, \frac{\lambda-\beta\lambda}{1+\lambda+\lambda^2}, -\frac{(-1+\beta)\lambda^2}{1+\lambda+\lambda^2}, 0\right\}$$

```
(*AND delay coordinate MA-coeficients for stock*)
Simplify[Coefficient[Expand[Numerator[F[[5]]][[1]]], {e^(-i w), e^(-2 i w), e^(-3 i w), e^(-4 i w)}]]
(*These are the α (t), ..., α (t-3) coeficients*)
```

$$\left\{\frac{1-\beta}{1+\lambda+\lambda^2}, \frac{\lambda-\beta\lambda}{1+\lambda+\lambda^2}, -\frac{(-1+\beta)\lambda^2}{1+\lambda+\lambda^2}, 0\right\}$$

```
(*A PLOT OF NORMALIZED SPECTRAL DENSITIES:*)
β = .4; γ = .3; λ = 0.4;
PSXN = PS[[1]][[1]] / NIntegrate[PS[[1]][[1]], {w, 0, Pi}];
PSYN = PS[[2]][[1]] / NIntegrate[PS[[2]][[1]] , {w, 0, Pi}];
PSSN = PS[[5]][[1]] / NIntegrate[PS[[5]][[1]], {w, 0, Pi}];
p1 = Plot[{PSXN, PSYN, PSSN}, {w, 0, Pi},
        DefaultFont -> {"Times-New-Roman", 12},
        PlotStyle -> {Thickness[0.007],
     {Thickness[0.007], Dashing[{0.03, 0.03}]}, {Thickness[0.015]}},
        AxesStyle -> {Thickness[0.007]},
        AxesLabel -> {FontForm["Frequency", {"Times-New-Roman", 12}],
       FontForm["Power", {"Times-New-Roman", 12}]},
        Ticks -> {{0, 3}, {0, 1}},
        AspectRatio -> 1,
        PlotRange → {0, 1.5}];
(* This is Fig 4 B in Bjornstad et al. -- Note though That the X-
  axis is in angular frequiencies (0 - π) *)
Clear[β, γ, λ]
```



```
(*Calculating the limit of the Transfer functions as d → Infinity
   (cf. Table 2)*)
d = Infinity;

(*defining the infinit denomators of the transfer functions*)
denominator = (1 + Exp[-I w] γ -
    (1 - β) Exp[- I w] (Sum[λ^(n - 1) Exp[-n I w], {n, 1, (d + 1)}] / Sum[λ^n, {n, 0, d}]));

(* 0 gr transfer function in the limit as d → Infinity*)
F0 = (1 + Exp[- I w] γ) / denominator
```

$$\frac{1 + e^{-i w} \gamma}{1 + e^{-i w} \gamma - \frac{e^{-i w} (1-\beta) (1-\lambda)}{e^{i w} - \lambda}}$$

```
(* k (> 0) gr transfer function in the limit as d → Infinity*)
Fk = (1 - β) Exp[-I k w] / denominator
```

$$\frac{e^{-i k w} (1 - \beta)}{1 + e^{-i w} \gamma - \frac{e^{-i w} (1-\beta) (1-\lambda)}{e^{i w} - \lambda}}$$

```
(* Stock  transfer function in the limit as d → Infinity*)
Fs = (Exp[- I w] (1 - β) (- Sum[λ^(n - 1) Exp[-n I w], {n, 1, (d + 1)}] / Sum[λ^n, {n, 0, d}])) /
  denominator
```

$$-\frac{e^{-i w} (1 - \beta) (1 - \lambda)}{\left(1 + e^{-i w} \gamma - \frac{e^{-i w} (1-\beta) (1-\lambda)}{e^{i w} - \lambda}\right) (e^{i w} - \lambda)}$$

```
(*A PLOT OF NORMALIZED SPECTRAL DENSITIES
  FOR THE LIMITING CASE: O-group and spawning stock*)
β = .4; γ = .3; λ = 0.4;
T0 = ExpToTrig[Simplify[F0]];
TS = ExpToTrig[Simplify[Fs]];
PS0 = Re[T0]^2 + Im[T0]^2;
PSS = Re[TS]^2 + Im[TS]^2;
PS0N = PS0 / NIntegrate[PS0, {w, 0, Pi}];
PSSN = PSS / NIntegrate[PSS, {w, 0, Pi}];
p1 = Plot[{PS0N, PSSN}, {w, 0, Pi},
        DefaultFont -> {"Times-New-Roman", 12},
        PlotStyle -> {Thickness[0.007],
     {Thickness[0.007], Dashing[{0.03, 0.03}]}, {Thickness[0.015]}},
        AxesStyle -> {Thickness[0.007]},
        AxesLabel -> {FontForm["Frequency", {"Times-New-Roman", 12}],
       FontForm["Power", {"Times-New-Roman", 12}]},
        Ticks -> {{0, 3}, {0, 1}},
        AspectRatio -> 1,
       PlotRange → {0, 1.5}];
(* Note That the X-axis is in angular frequiencies (0 - π) *)
Clear[β, γ, λ]
```